libname miR 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR';

libname indR 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\independents\indR';
libname demR 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\democrats\demR';
libname repR 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\republicans\repR';

data miR.all; set indR.all demR.all repR.all; run;

proc freq data=miR.all; table partyid forcedparty; run;


proc freq data=miR.all; tables condition conditionnum; run;

proc freq data=indR.all; tables condition; run;
proc freq data=demR.all; tables condition; run;
proc freq data=repR.all; tables condition; run;

**********************************CLEANING DATA**************************************;

*coding edu as continuous;
data miR.all; set miR.all;
format education 1.;
if edu = 1 then education = 1; *elementary school;
else if edu = 2 then education = 2; *junior high;
else if edu = 3 then education = 3; *some high school;
else if edu = 4 then education = 4; *high school graduate;
else if edu in (5,6) then education = 5; *some college and associate's degree;
else if edu = 7 then education = 6; *bachelor's degree;
else if edu in (8,9,14) then education = 7; *some graduate school, master's degree, MBA;
else if edu in (10,11,13) then education = 8; *JD, MD, other advanced degree;
else if edu = 12 then education = 9; run; *PhD;

*getting rid of non-US citizens - asked for this, but didn't get it;
data miR.all; set miR.all; 
if citizen NE "us" then delete; run;
*120 subjects deleted;

TITLE 'Completers vs Noncompleters';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Completers vs Noncompleters.html';

*how are consenters different from completers of the DV?;
data miR.all; set miR.all; 
if consented = . then delete; *159 subjects deleted; **same;
if debriefed = . then debriefed = 0; run;
proc ttest data=miR.all; class debriefed; var age edu; run;
proc freq data=miR.all; 
tables gender*debriefed  / chisq ; run;

ODS HTML CLOSE;

*getting rid of people who did not complete the 2 main DVs;
data miR.all; set miR.all; 
if planiat = . then delete; 
if planprefer = . then delete; run;
*1246 subjects deleted;


*recoding variables;
data miR.all; set miR.all; 
socialpolitics = .;
if socpolitics = 1 then socialpolitics = 3; 
else if socpolitics = 2 then socialpolitics = 2; 
else if socpolitics = 3 then socialpolitics = 1; 
else if socpolitics = 4 then socialpolitics = 0; 
else if socpolitics = 5 then socialpolitics = -1; 
else if socpolitics = 6 then socialpolitics = -2; 
else if socpolitics = 7 then socialpolitics = -3; run;

data miR.all; set miR.all; 
economicpolitics = .;
if econpolitics = 1 then economicpolitics = 3; 
else if econpolitics = 2 then economicpolitics = 2; 
else if econpolitics = 3 then economicpolitics = 1; 
else if econpolitics = 4 then economicpolitics = 0; 
else if econpolitics = 5 then economicpolitics = -1; 
else if econpolitics = 6 then economicpolitics = -2; 
else if econpolitics = 7 then economicpolitics = -3; run;

data miR.all; set miR.all; 
partyforce = .;
if forcedparty = 1 then partyforce = -3; 
else if forcedparty = 2 then partyforce = -2; 
else if forcedparty = 3 then partyforce = -1; 
else if forcedparty = 4 then partyforce = 0; 
else if forcedparty = 5 then partyforce = 1; 
else if forcedparty = 6 then partyforce = 2; 
else if forcedparty = 7 then partyforce = 3; run;

data miR.all; set miR.all; 
informat majorparty $9.;
majorparty = "."; 
if partyid = 1 then majorparty = "Dem"; 
else if partyid = 2 then majorparty = "Rep"; 
else if partyid = 3 then majorparty = "Ind"; run;

data miR.all; set miR.all; 
informat indforce $9.;
indforce = "."; 
if partyforce in (1,2,3) then indforce = "DemInd"; 
else if partyforce = 0 then indforce = "StrongInd"; 
else if partyforce in (-1,-2,-3) then indforce = "RepInd"; run;

data miR.all; set miR.all; 
detailsinfluenced = howdetails; run;

data miR.all; set miR.all; 
preferumb = .;
if planprefer = 1 then preferumb = 3; 
else if planprefer = 2 then preferumb = 2; 
else if planprefer = 3 then preferumb = 1; 
else if planprefer = 4 then preferumb = 0; 
else if planprefer = 5 then preferumb = -1; 
else if planprefer = 6 then preferumb = -2; 
else if planprefer = 7 then preferumb = -3; run;

proc means data=miR.all; var preferumb planprefer; run;
proc corr data=miR.all; var preferumb planprefer; run;

data miR.all; set miR.all; 
poliat = 0 - politicsiat; run;

data miR.all; set miR.all; 
informat poliatbin $19.;
poliatbin = "."; 
if poliat > .15 then poliatbin = "Implicit Democrat";
else if poliat < -.15 then poliatbin = "Implicit Republican"; run;

data miR.all; set miR.all; 
conditionnum = .; 
if condition = "dem" then conditionnum = .5; 
else if condition = "rep" then conditionnum = -.5; run;

data miR.all; set miR.all; 
if partyinfluenced = 1 then aware = -.5; 
else if partyinfluenced in (2,3,4,5) then aware = .5; run;



************************************************************************
CLEAN DATASET INCLUDES 1865 who:
only us citizens 
only people who completed both 2 DVs 

RETAINED PEOPLE who:
failed the IMC
************************************************************************;

**********************************Sample Characteristics**************************************;

TITLE 'Combined Sample Characteristics';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Combined Sample Characteristics.html';

proc freq data=miR.all;
tables gender racecen ethniccen citizen partyid vote2008; run;

proc univariate data=miR.all; var education edu; run;
proc freq data=miR.all; table education edu; run;

proc means data=miR.all; 
var relID socialpolitics economicpolitics poliat age; run;

proc sort data=miR.all; by majorparty;

proc freq data=miR.all; by majorparty; 
tables gender racecen ethniccen citizen partyid vote2008 edu; run;

proc means data=miR.all; by majorparty;
var relID socialpolitics economicpolitics poliat age; run;
*biasinfluence - higher is more agreement with decisions can be biased 1 to 7;

proc freq data=miR.all; by majorparty; 
table partyforce indforce; run;

ODS HTML CLOSE;


*************************making new datasets to analyze data easily*********************;
data miR.ind; set miR.all; 
keep = 0; 
if majorparty = "Ind" then keep = 1; 
if keep = 0 then delete; run;

data miR.strong; set miR.ind; 
keep = 0; 
if partyforce = 0 then keep = 1; 
if keep = 0 then delete; run;

data miR.rep; set miR.all; 
keep = 0; 
if majorparty = "Rep" then keep = 1; 
if keep = 0 then delete; run;

data miR.dem; set miR.all; 
keep = 0; 
if majorparty = "Dem" then keep = 1; 
if keep = 0 then delete; run;


*************manipulation checks****************;

TITLE 'Manip Checks';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Manip Checks.html';

proc freq data=miR.all; tables IMC; run;

proc sort data=miR.all; by condition;
proc freq data=miR.all; by condition;
tables manip1 manip2; run; 

proc freq data=miR.all; table manip3 manip4 howcarefully; run; 

proc sort data=miR.all; by IMC condition;
proc freq data=miR.all; by IMC condition;
tables manip1 manip2; run;

proc sort data=miR.all; by IMC;
proc freq data=miR.all; by IMC;
tables manip3 manip4 howcarefully; run;

ODS HTML CLOSE;

**************************************DESCRIPTIVES*********************************************;

TITLE 'Combined Sample Analysis';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Combined Analysis.html';

proc means data=miR.all ; 
var preferumb planiat; run;

proc corr data=miR.all; var planiat preferumb; run;

proc means data=miR.all ; class majorparty;
var preferumb planiat; run;

proc sort data=miR.all; by majorparty; 
proc corr data=miR.all;  by majorparty; var planiat preferumb; run;

proc means data=miR.all ; class majorparty;
var welfareexperience knowledgewelfare partyinfluenced 
socialinfluenced detailsinfluenced biasinfluence; run;
*higher is more experience, influence 1 to 5;

proc means data=miR.all; 
var welfareexperience knowledgewelfare partyinfluenced socialinfluenced detailsinfluenced; run;
*higher is more experience, influence 1 to 5;

*looking at condition/party interactions;
proc sort data=miR.all; by majorparty; 
proc freq data=miR.all ; by majorparty;
tables partyinfluenced detailsinfluenced; run;
proc means data=miR.all ; class majorparty;
var preferumb planiat; run;

proc means data=miR.all ; by majorparty; class condition; 
var planiat preferumb; run;

*ideal combined graph would have dems, reps, ind, and then implicit rep and dem ind, so 5 bars;
proc means data=miR.all; by majorparty; class poliatbin condition; 
var planiat preferumb; run;

*testing hypotheses on reasoning;
proc glm data=miR.all ; class majorparty; 
model partyinfluenced=majorparty; means majorparty / tukey; run;

ODS HTML CLOSE;


/*************************************************************************************************************************************
**************analysis with manipulation check failures removed on different datasets*************************************************
*************************************************************************************************************************************/

TITLE 'Main Analysis with Study Manip Check Failures Removed';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Main Analysis with Study Manip Check Failures Removed.html';

data miR.manip; set miR.all; run;

data miR.manip; set miR.manip;
if condition = "rep" AND manip1 = 1 then delete; run;
*deleted 125 for seeing Dems propose CAP but reporting that Reps did;

data miR.manip; set miR.manip;
if condition = "rep" AND manip2 = 2 then delete; run;
*deleted 15 (not deleted above) for seeing Reps propose UAP but reporting Dems did;

data miR.manip; set miR.manip;
if condition = "dem" AND manip1 = 2 then delete; run;
*deleted 56 for seeing Reps propose CAP but reporting that Dems did;

data miR.manip; set miR.manip;
if condition = "dem" AND manip2 = 1 then delete; run;
*deleted 8 (not deleted above)for seeing Dems propose UAP but reporting that Reps did;

data miR.manip; set miR.manip;
if manip3 = 1 then delete; run; 
*deleted 35 (not deleted above) for saying that CAP was the more generous plan; 

*in the paper, i report the percentage of people who failed the study specific manipulation checks
as the sum of the people above divided by 1381, which is the number of people who completed manipcheck 3 
(1380 completed manips1 and 2);

proc freq data=miR.IMC; table manip1 manip2 manip3; run;

proc sort data=miR.manip; by majorparty; 

proc ttest data=miR.manip; by majorparty; 
class condition; var preferumb planiat; run;

proc glm data=miR.manip (where=(majorparty="Ind")); 
class condition;
model preferumb=condition|poliat; run;

proc glm data=miR.manip (where=(majorparty="Ind")); 
class condition;
model planiat=condition|poliat; run;

proc glm data=miR.manip (where=(indforce="StrongInd")); 
class condition;
model preferumb=condition|poliat; run;

proc glm data=miR.manip (where=(indforce="StrongInd")); 
class condition;
model planiat=condition|poliat; run;

proc sort data=miR.manip; by majorparty; 

proc means data=miR.manip ; by majorparty; 
class poliatbin condition; 
var planiat preferumb; run;

ODS HTML CLOSE;


TITLE 'Main Analysis with IMC Failures Removed';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Main Analysis with IMC Failures Removed.html';

data miR.IMC; set miR.all;
if IMC = 0 then delete; run;
*deleted 471;

*here i knew how many people actually complete the IMC, so i used that percentage;

proc freq data=miR.IMC; table IMC; run; 

proc sort data=miR.IMC; by majorparty; 

proc ttest data=miR.IMC; by majorparty; 
class condition; var preferumb planiat; run;

proc glm data=miR.IMC (where=(majorparty="Ind")); 
class condition;
model preferumb=condition|poliat; run;

proc glm data=miR.IMC (where=(majorparty="Ind")); 
class condition;
model planiat=condition|poliat; run;

proc glm data=miR.IMC (where=(indforce="StrongInd")); 
class condition;
model preferumb=condition|poliat; run;

proc glm data=miR.IMC (where=(indforce="StrongInd")); 
class condition;
model planiat=condition|poliat; run;

proc sort data=miR.IMC; by majorparty; 

proc means data=miR.IMC ; by majorparty; 
class poliatbin condition; 
var planiat preferumb; run;

ODS HTML CLOSE;




/*************************************************************************************************************************************
**********************************************************Analyses********************************************************************
*************************************************************************************************************************************/


TITLE 'Rep-Dem Average Difference';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Rep-Dem Average Difference.html';
data miR.two; set miR.all; 
keep = 0; 
if majorparty in ("Dem", "Rep") then keep=1; 
if keep = 0 then delete; run;

proc ttest data=miR.two; class majorparty; var preferumb planiat; run;
ODS HTML CLOSE; 


**********************************DEMOCRATS**************************************;

TITLE 'Democrats Analysis';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Democrats Analysis.html';

*test of hypothesis;
proc ttest data=miR.dem; 
class condition; var preferumb planiat; run;

*covarying DV order;
proc glm data=miR.dem ; class condition dvorder;
model preferumb=condition|dvorder; run;

proc glm data=miR.dem; class condition dvorder;
model planiat=condition|dvorder; run;

proc sort data=miR.dem ; by condition; 
proc corr data=miR.dem ; by condition; 
var planiat preferumb detailsinfluenced partyinfluenced; run;

ODS HTML CLOSE;


**********************************REPUBLICANS**************************************;

TITLE 'Republicans Analysis';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Republicans Analysis.html';

*test of hypothesis;
proc ttest data=miR.rep ; 
class condition; var preferumb planiat; run;

*covarying DV order;
proc glm data=miR.rep; class condition dvorder;
model preferumb=condition|dvorder; run;

proc glm data=miR.rep ; class condition dvorder;
model planiat=condition|dvorder; run;

proc sort data=miR.rep ; by condition; 
proc corr data=miR.rep ; by condition; 
var planiat preferumb detailsinfluenced partyinfluenced; run;

ODS HTML CLOSE;


**********************************INDEPENDENTS**************************************;

TITLE 'CIs on main analyses';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\CIs on main analyses.html';

proc glm data=miR.ind; 
model preferumb=conditionnum|poliat / CLPARM; run;

proc glm data=miR.ind; 
model planiat=conditionnum|poliat / CLPARM; run;

proc glm data=miR.strong;
model preferumb=conditionnum|poliat / CLPARM; run;

proc ttest data=miR.strong; class conditionnum; var preferumb; run;

proc glm data=miR.strong;
model planiat=conditionnum|poliat / CLPARM; run;

data miR.ind; set miR.ind; 
informat extremist 1.; 

if econpolextreme in (1,2) then extremist = .5;
else if socialpolextreme in (1,2) then extremist = .5;

if econpolextreme = 3 then do; 
extremist = -.5;
if socialpolextreme in (1,2) then extremist = .5; end; 

if socialpolextreme = 3 then do;
extremist = -.5;
if econpolextreme in (1,2) then extremist = .5; end; 
run;

proc freq data=miR.ind; table extremist; run;

proc glm data=miR.ind; 
model preferumb=conditionnum|poliat|extremist / CLPARM; run;

proc glm data=miR.ind; 
model planiat=conditionnum|poliat|extremist / CLPARM; run;

ODS HTML CLOSE; 



TITLE 'Independents - IAT incremental validity WITH CIs';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Independents - IAT incremental validity WITH CIs.html';

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce / CLPARM ; run;

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce poliat / CLPARM ; run;

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce poliat
partyinfluenced*conditionnum partyforce*conditionnum / CLPARM ; run;

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce poliat
partyinfluenced*conditionnum partyforce*conditionnum poliat*conditionnum / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce poliat / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce poliat
partyinfluenced*conditionnum partyforce*conditionnum / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce poliat
partyinfluenced*conditionnum partyforce*conditionnum poliat*conditionnum / CLPARM ; run;

ODS HTML CLOSE;



TITLE 'Independents - IAT incremental validity WITH CIs WITH political ideology';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Independents - IAT incremental validity WITH CIs WITH political ideology.html';

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce politics / CLPARM ; run;

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce politics poliat / CLPARM ; run;

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce politics poliat
partyinfluenced*conditionnum partyforce*conditionnum politics*conditionnum / CLPARM ; run;

proc glm data=miR.ind; 
model preferumb=conditionnum partyinfluenced partyforce politics poliat
partyinfluenced*conditionnum partyforce*conditionnum politics*conditionnum poliat*conditionnum / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce politics / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce politics poliat / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce politics poliat
partyinfluenced*conditionnum partyforce*conditionnum politics*conditionnum / CLPARM ; run;

proc glm data=miR.ind; 
model planiat=conditionnum partyinfluenced partyforce politics poliat
partyinfluenced*conditionnum partyforce*conditionnum politics*conditionnum poliat*conditionnum / CLPARM ; run;

ODS HTML CLOSE;


**********************************FOLLOW-UP ANALYSES**************************************;

TITLE 'Full Sample - IAT incremental validity WITH CIs';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Full Sample - IAT incremental validity WITH CIs.html';

data miR.major; set miR.all; 
keep = 0; 
if majorparty in ("Dem", "Rep", "Ind") then keep=1; 
if keep = 0 then delete; run;

data miR.major; set miR.major; 
informat newparty 1.; 
newparty = .; 
if majorparty = "Rep" then newparty = -2; 

if majorparty = "Ind" then do; 
if indforce = "RepInd" then newparty = -1; 
else if indforce = "StrongInd" then newparty = 0; 
else if indforce = "DemInd" then newparty = 1; end; 

else if majorparty = "Dem" then newparty = 2; run;

proc glm data=miR.major; 
model preferumb=conditionnum partyinfluenced newparty / CLPARM; run;

proc glm data=miR.major; 
model preferumb=conditionnum partyinfluenced newparty poliat / CLPARM; run;

proc glm data=miR.major; 
model preferumb=conditionnum partyinfluenced newparty poliat
partyinfluenced*conditionnum newparty*conditionnum / CLPARM; run;

proc glm data=miR.major; 
model preferumb=conditionnum partyinfluenced newparty poliat
partyinfluenced*conditionnum newparty*conditionnum poliat*conditionnum / CLPARM; run;

proc glm data=miR.major; 
model planiat=conditionnum partyinfluenced newparty / CLPARM; run;

proc glm data=miR.major; 
model planiat=conditionnum partyinfluenced newparty poliat / CLPARM; run;

proc glm data=miR.major; 
model planiat=conditionnum partyinfluenced newparty poliat
partyinfluenced*conditionnum newparty*conditionnum / CLPARM; run;

proc glm data=miR.major; 
model planiat=conditionnum partyinfluenced newparty poliat
partyinfluenced*conditionnum newparty*conditionnum poliat*conditionnum / CLPARM; run;

ODS HTML CLOSE; 



TITLE1 'Awareness';
ODS HTML BODY = 'C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\results\Awareness.html';

proc sort data=miR.all; by majorparty; 
proc means data=miR.all n mean std; by majorparty; class condition aware; var preferumb planiat; run;

Title1 'Independents - Continuous Awareness';
proc glm data=miR.ind; 
model preferumb=conditionnum|poliat|partyinfluenced / CLPARM; run;

proc glm data=miR.ind; 
model planiat=conditionnum|poliat|partyinfluenced / CLPARM; run;

Title1 'Independents - Dichotomous Awareness';
proc glm data=miR.ind; 
model preferumb=conditionnum|poliat|aware / CLPARM; run;

proc glm data=miR.ind; 
model planiat=conditionnum|poliat|aware / CLPARM; run;

Title1 'Independents - Dichotomous Awareness 3-way removed';
proc glm data=miR.ind; 
model preferumb=conditionnum poliat aware
conditionnum*poliat conditionnum*aware/ CLPARM; run;

proc glm data=miR.ind; 
model planiat=conditionnum poliat aware
conditionnum*poliat conditionnum*aware/ CLPARM; run;

TITLE1 'Democrats - Dichotomous Awareness';
proc glm data=miR.dem; 
model preferumb=conditionnum|aware / CLPARM; run;

proc glm data=miR.dem; 
model planiat=conditionnum|aware / CLPARM; run;

TITLE1 'Republicans - Dichotomous Awareness';
proc glm data=miR.rep; 
model preferumb=conditionnum|aware / CLPARM; run;

proc glm data=miR.rep; 
model planiat=conditionnum|aware / CLPARM; run;

ODS HTML CLOSE; 



**********************************GRAPHING**************************************;

*reverse coding implicit party identity for graphing only - now ID with democrats is negative so that it is on the left for graphing purposes;
data miR.ind; set miR.ind; 
poliat_reverse = 0-poliat; run;
proc corr data=miR.ind; var poliat poliat_reverse; run;
proc means data=miR.ind; var poliat poliat_reverse preferumb planiat; run;


ods listing close;
goptions reset=all device=pdf hsize=4.6in vsize=3in gsfmode=replace gsfname=grafout HORIGIN=1 vorigin=7 ftext=ariel ;
ods pdf file='C:\Users\Carlee\Documents\Research Projects\Moderates\motivatedindependents\analysis\miR\graphs\study 1 graphs.pdf' startpage=never;

AXIS1 LABEL=(ANGLE=90 ROTATE=0 h=9pt "Explicit Preference for Generous Plan") MINOR=(n=1) order=-3.8 to 4 by 1.9 value=(h=9pt);
AXIS2 LABEL=(h=2pt "  ") minor=none order=-1 to .75 by .25 value=(h=9pt);
legend1 label=none shape=line(20pt) mode=share position=(top right inside) across=1 
value=( h=9pt j=l "Democrats Proposed Generous Plan" "Republicans Proposed Generous Plan");
SYMBOL1 INTERPOL=R C=black line=1 mode=include w=3;
SYMBOL2 INTERPOL=R C=black line=2 mode=include w=3; 
PROC GPLOT DATA=miR.ind ; 
	PLOT	preferumb*poliat_reverse=condition / vaxis=axis1 haxis=axis2 legend=legend1 noframe ; RUN;quit;

goptions reset=all device=pdf hsize=4.6in vsize=3in gsfmode=replace gsfname=grafout HORIGIN=1 vorigin=4 ftext=ariel;
AXIS1 LABEL=(ANGLE=90 ROTATE=0 h=9pt "Implicit Preference for Generous Plan") MINOR=(n=1) order=-.7 to 1.1 by .35 value=(h=9pt);
AXIS2 LABEL=(h=9pt "Implicitly Democratic                 Implicitly Republican") minor=none order=-1 to .75 by .25 value=(h=9pt);
SYMBOL1 INTERPOL=R C=black line=1 mode=include w=3;
SYMBOL2 INTERPOL=R C=black line=2 mode=include w=3; 
PROC GPLOT DATA=miR.ind ; 
	PLOT	planiat*poliat_reverse=condition / vaxis=axis1 haxis=axis2 nolegend noframe ; RUN;quit;


ods pdf close;
ods listing;


*removing selected demographics that could possibly be identifying for posting data;
proc contents data=miR.all; run;

data miR.post; set miR.all; 
drop byear bmonth dayofbirth characteristic genoccupation major reldenom  relfamily religion work  zip; run;

proc contents data=miR.post; run;
